

3-Axis Accelerometer

Tilt, shake and which way is up, from an SC7A20H at 0x19 on two shared wires.




3V3 means 3.3 V

The chip runs from 1.71 to 3.6 V and **3.6 V is its absolute maximum**. The board has no regulator, so what goes into 3V3 is what the chip gets. Never 5V. A 5 V Arduino Uno needs a level converter (TK97) between its SDA and SCL and this board.

Pins, and which to wire

Chip side up, header at the bottom. All four pins are wired. The red light only says 3V3 arrived: the chip answering at 0x19 is the sign it works.



A still board reads 1 g, not 0

It measures the push holding it up. Flat, chip side up: X 0, Y 0, Z **+1 g**. X points down the board to the header, Y across to the SDA end, Z out of the chip side; an axis reads +1 g pointing up. Only a falling board reads 0.

Three numbers to read it by

Resolution

1 mg a count
at ± 2 g

12 bits. ± 4 , ± 8 and ± 16 g reach further at 2, 4 and 8 mg a count.

Zero-g offset

± 90 mg typical,
mounted

About 5° of tilt; 120 mg at most. Measure it level once and subtract it.

Current

16 μ A at 100
readings a second

0.5 μ A asleep. The power light draws about 1.3 mA.

Tilt is only as good as the zero: a mounted chip's typical 90 mg offset is about 5° , and it does not average away. Level the board once, note what it reads, and subtract that from every reading after.

Typical values from Silan's SC7A20H data sheet, version 0.7; the angle is the arcsine of the offset, and the light's current is Ohm's law on the 1 kΩ with a red LED's usual 2.0 V. Not measured on a bench.

The first reading, five a second

Arduino IDE · Board: your board, e.g. ESP32S3 Dev Module · USB CDC On Boot: Enabled (S3 only) · No library: Wire only

```
#include <Wire.h>

const uint8_t ACCEL_ADDR = 0x19; // SDO open on the board

void writeReg(uint8_t reg, uint8_t v) {
  Wire.beginTransmission(ACCEL_ADDR);
  Wire.write(reg); Wire.write(v);
  Wire.endTransmission();
}

bool readRegs(uint8_t reg, uint8_t *b, uint8_t n) {
  Wire.beginTransmission(ACCEL_ADDR);
  Wire.write(reg | 0x80); // step through registers
  if (Wire.endTransmission(false) != 0) return false;
  if (Wire.requestFrom(ACCEL_ADDR, n) != n) return false;
  for (uint8_t i = 0; i < n; i++) b[i] = Wire.read();
  return true;
}

void setup() {
  Serial.begin(115200);
  while (!Serial) delay(10); // native USB
  Wire.begin();
  uint8_t id = 0; // WHO_AM_I
  if (!readRegs(0x0F, &id, 1) || id != 0x11) {
    Serial.println("no SC7A20: check GND, SDA, SCL");
    while (true) delay(100);
  }
  writeReg(0x20, 0x57); // wake: 100 Hz, X Y Z
  writeReg(0x23, 0x80); // +-2 g
}

void loop() {
  uint8_t b[6];
  if (readRegs(0x28, b, 6)) {
    for (int i = 0; i < 3; i++) {
      int16_t c = (int16_t)(b[2 * i + 1] << 8 | b[2 * i]) >> 4;
      Serial.print(c / 1000.0, 3); // 1 mg a count
      Serial.print(i < 2 ? " " : " g\n");
    }
  }
  delay(200);
}
```

0.031 -0.012 1.004 g
0.028 -0.016 1.008 g
Flat, chip side up: Z near 1. Stood on its header, X near -1.

Tilt and shake, from the same 1 g

Tilt: held still, $\text{pitch} = \text{atan2}(-x, \sqrt{y^2 + z^2})$ and $\text{roll} = \text{atan2}(y, z)$, in radians; pitch rises as the header end goes down. Trust the angle only while the total, $\sqrt{x^2 + y^2 + z^2}$, is within 0.1 g of 1. **Shake**: anything more than 0.5 g from 1 in total is a jolt. Write 0x77 to 0x20 for 400 readings a second, or a tap falls between two.

When it does not work

No SC7A20 at 0x19

Count from the square pad: GND, 3V3, SCL, SDA. SDA and SCL swapped is the usual fault.

A LIS3DH library finds nothing

It looks at 0x18 for a name of 0x33. This chip is 0x19 and 0x11: use Wire.

Z reads -1

The chip side faces down. An axis reads +1 g pointing up, -1 g pointing down.

Specifications

Chip	Silan SC7A20H, LGA-12, 2 × 2 mm	Address	0x19, fixed; WHO_AM_I 0x11
Range	±2, ±4, ±8 or ±16 g	Output	12 bits, 1 mg a count at ±2 g
Data rate	1.56 to 4434 Hz; off at power-up	Supply	1.71 to 3.6 V: use 3.3 V
Pull-ups	4.7 kΩ to 3V3, always on	Board	22.4 × 30.4 mm, header fitted

Wiring

	FRONT	S3	ESP32	PICO	UNO VIA TK97
1	GND	GND	GND	GND	GND
2	3V3	3V3	3V3	3V3	TK97 3V3
3	SCL	GPIO 9	GPIO 22	GP5	A5
4	SDA	GPIO 8	GPIO 21	GP4	A4

Hole 1 is the square pad. 3V3 is the chip's supply: never 5V. The SDA and SCL pins are each board's defaults, so Wire.begin() needs no pin numbers. On an Uno, go through a TK97: the Uno's 5V feeds the TK97, its 3V3 output feeds this board.

Every reading is 0.000

Still asleep: 0x20 was never written. Write 0x57 to wake it at 100 Hz.

Level, but X reads 0.08

The mounted offset, up to 120 mg. Measure it level once and subtract it.

Two boards, one shows up

Both are at 0x19 and it cannot change. Use the second I²C bus of an ESP32.